
News
Spike2 and Signal Training Days

Our U.K. Spike2 and Signal training courses will take place on Thursday 27" and Friday
28" of March at St. John'’s Innovation Centre, Cambridge. We have now been able to
secure additional room at the Innovation Centre and have re-opened bookings for the
Spike2 sessions for a limited time only. There are still places available for Signal users
and those interested in attending the script or workshop sessions.

If you or any of your colleagues would like to attend, please follow the links to view the
training day program and complete the registration form. Please give a short description
of the type of research that you do and analysis that you are interested in. This will allow
us to tailor sessions for specific requirements.

New distributor for China

We are pleased to announce that Shanghai Qichi Instruments Co. Ltd. are acting as our
distributor for mainland China. For contact details please click on the link above.

Meetings and events

Physiological Society Focused Meeting: Cardiac & Respiratory Physiology
University of Leeds

Leeds, UK

March 17" — 19" 2008

Experimental Biology 2008
Booth 244

San Diego, California

April 5™ — 8™ 2008

Chinese Educational Conference of Physiology
Guilin City

Guangxi Province, China

April 10" — 13" 2008

18th Annual Meeting of the Society for the Neural Control of Movement
Naples Beach Hotel

Naples, Florida

April 29" — May 4™ 2008

Vision Sciences Society Annual Meeting
Naples Grande Hotel

Naples, Florida

May 9" — 14" 2008

Established 1970

http://www.ced.co.uk/nwu.shtml?nw9u.htm
http://www.ced.co.uk/nwu.shtml?tdapplu.htm
http://www.qichi-instruments.com/
http://www.physoc.org/meetings/leeds2008/
http://www.eb2008.org/
http://ncm-society.org/index.php?page=annual_meeting
http://www.visionsciences.org/meeting.html

Latest versions of Spike2 and Signal

Updates for Spike2 and Signal are available from the CED downloads page, or by
clicking on the links in the table below. Demonstration versions of the latest software are

also available.

Spike2 downloads

Signal downloads

Spike2 version 6.07

Signal version 4.00

Spike2 version 5.19

Signal version 3.10

Cloze

Spike2 demo Signal demo
Q. Can | display a power spectrum result with log axes?
A. You can choose to display data in result and XY views using logarithmic axes.
Both the X and Y axis dialogs have checkboxes to enable the log axes display.

Channel |NNESERES] Optimise |
Top  |0.02475851 33| | Showal |
Boltom 24788512006 19|
I™ Large tick spacing « [10
I Tick subdivisions E= [ e

¥ Logarithmic W Show Powers

Y-range dialog

With the logarithmic checkbox enabled, the large tick spacing field controls the
multiplying factor between large ticks and is set to 10 by default. The Show
Powers checkbox is also enabled and will display the axis labels at large tick
marks as powers of 10 if checked.

Log axes display on the X and Y range

From the Display tab of the Edit preferences dialog you can also set the number
of decades to display by default when you select Logarithmic mode in the X and
Y-axis dialogs.

http://www.ced.co.uk/uplsu.shtml
http://www.ced.co.uk/s2wu6u.shtml
http://www.ced.co.uk/s2wu5u.shtml
http://www.ced.co.uk/s2demu.shtml
http://www.ced.co.uk/sigw4u.shtml
http://www.ced.co.uk/sigw3u.shtml
http://www.ced.co.uk/sigdemu.shtml
CED
The grid display option shown here can be enabled in the Show/Hide channels dialog


Q.

>

Edit preferences dialog

Signal

—

How do | set a waveform channel from a given frame to be used for arbitrary
waveform output in the pulse dialog?

You can copy and paste frame data in Signal simply by using the Edit menu
Copy and Paste commands or the standard keyboard shortcuts of CtrI+C for
copy and Ctrl+V for paste. The data copied to the clipboard is that which is
displayed in the selected window. In this way data can be copied and pasted
between data files, result views and XY views as well as used for arbitrary
waveform output.

You can set up an arbitrary waveform play area in the pulses dialog by clicking
and dragging the arbitrary waveform output icon from the palette and dropping it
on an enabled DAC output to create an output wave area as seen below. With
the area selected you can paste data from the clipboard using Ctr1+V or right-
clicking in the play area. You can match the output play rate to the original
sampling rate used to acquire the data from the Paste waveform dialog.

Pre-recorded action potential data copied for waveform output

Scripts: Spike2

Q.

Is there any way to test on-line scripts on existing data files? It is quite time
consuming to set-up and run an experiment each time | want to test the on-line
script that | am currently working on so it would be very useful to be able to use a
re-running data file for testing purposes.

The attached script, OnlineTest.s2s, is an example ‘skeleton’ that can be
used to test on-line control and analysis functions on an existing data file. The
script contains an idle routine which replays a selected channel in a data file and

Back to top

Right-click the script icon
and save to disk.

If you have any problems
opening the embedded
scripts in this newsletter
please let us know.



'$OnlineTest|Example 'skeleton' script to test on-line functions using an existing channel in a data file

'It contains a toolbar to select an open data file to replay and

'start and stop buttons to fake sampling in realtime. It displays the log view at the bottom

'of the screen to place values in that the user may wish to extract.

'Hides/restores unwanted windows.



'The script is a 'work in progress' and is offered without guarantees.

'You must test it to see whether it is suitable for your application.



'CED 10/03/08



var data%;								'Handle of selected data file

var sTime;								'The last time we looked at the idle routine

var chans%[32];

var playChan%;

var ok%;



HideAll();                       'Hide all unwanted bars

ToolbarVisible(1);					'Make toolbar visible always

Select%();									'Set up file to use

DoToolbar();							'Do the toolbar

RestoreAll();                    'Restore all bars

Halt;



func Select%()								'New sampling window

View(LogHandle());              'Make log view the current view

EditSelectAll();                'Select all text in log view

EditClear();                    'Delete it

Window(0,80,100,100);           'Display it at the bottom of the screen

WindowVisible(1);               'Make it visible



if data%>0 then						'If there is already a data view open...

  View(data%);							'...close it

  FileClose();                   'No point setting data% to 0...

endif;



data%:=FileOpen("",0);		'Open a new data file for sampling - For on-line replace FileOpen() with FileNew()

if data%<0 then Message("Unable to open new data file");Halt() endif;



FrontView(data%);						'Bring the data view to the front



Window(0,0,100,80);					'Make data window in top bit of screen

XRange(0,10);

ChanList(chans%[],1+16+512);        'Get all existing compatible channels

ChanShow(chans%[]);



DlgCreate("Select Channel");

DlgChan(1,"Set channel to use for replay", 1+16+512);       'Set channel to use for replay

ok%:= DlgShow(playChan%);



if ok% < 1 then

    Message("Script cancelled");        'Quit script if user cancels

    halt;endif;



ChanHide(-1);

ChanShow(playChan%);        'Display the selected channel only



ToolbarEnable(3,0);					'Disable "Sample stop" button

ToolbarEnable(2,1);					'Disable "Sample stop" button

ToolbarText("Press REPLAY START to commence sampling");

return 1;

end;



proc DoToolbar()



ToolbarSet(1,"&Quit",Quit%);		'Set up toolbar buttons

ToolbarSet(2,"&Replay start", Start%);

ToolbarSet(3,"&Replay stop", Stop%);

ToolbarSet(4,"&Select file", Select%);

ToolbarEnable(3,0);					'Disable "Sample stop" button

Toolbar("Press REPLAY START to commence sampling", 1023);'Wait here until quit

end;



func Quit%()							'If "Quit" is pressed

SampleStop();							'Stop sampling

return 0;								'leave toolbar

end;



func Start%()							'If "Start" is pressed

sTime := 0;                             'set start of anaylsis time

PlayOffline(playChan%,-1,0,MaxTime(),1,1,2);    'Do offline output to simulate sampling (If using in on-line idle simply replace this with SampleStart())

ToolbarSet(0,"",Idle%);				'Call Idle%() whenever there is free time

ToolbarEnable(4,0);					'Disable "New file" button

ToolbarEnable(3,1);					'Enable "Sample stop" button

ToolbarEnable(2,0);					'Disable "Sample start" button

ToolbarEnable(1,0);					'Disable "Quit" button

ToolbarText("Press REPLAY STOP to stop sampling");

return 1;								'Stay with toolbar

end;



func Stop%()							'If "Stop" is pressed

ToolbarClear(0);

PlayOffLine(-1);                'Stop offline output (If using on-line simply replace this with SampleStop())

if (data%>0) and                 'if no data in file, stop will close it

   (ViewKind(data%) <> 0) then data% := 0 endif;

ToolbarEnable(4,1);					'Enable "New file" button

ToolbarEnable(3,0);					'Disable "Sample stop" button

ToolbarEnable(1,1);					'Enable "Quit" button

ToolbarText("Press SELECT FILE to replay another file");

return 1;								'Stay in toolbar

end;



func Idle%()							'Idle routine runs when system has time

var eTime;

if ViewKind(data%)=0 then        'This checks that the data view exists...

   View(data%);                  '...as this gives an error if it doesn't

    

	eTime := PlayOffline(0);                                        'Get current time during rerun (For on-line use, replace with MaxTime())

    

	if (eTime > sTime) then

        'PLACE CODE IN HERE TO ANALYSE THE DATA FILE BETWEEN sTime and eTime

      sTime:=eTime;

      endif;

   endif;

return 1;								'Stay in toolbar

end;



'You can find this code in the App() on-line help.

var gFloat%[20];              'global for floating window states

proc HideAll()

var i%;

gFloat%[0] := App(-2);        'number of windows

for i% := 1 to gFloat%[0] do  'hide all windows and save state

   gFloat%[i%] := View(App(i%)).WindowVisible(0);

   next;

end



proc RestoreAll()

var i%;

for i% := 1 to gFloat%[0] do  'restore hidden windows

   View(App(i%)).WindowVisible(gFloat%[i%]);

   next;

end







CED
OnlineTest.s2s

CED
You can set colour maps to use for the sonogram draw mode in the Display tab of the Edit preferences dialog

CED
You can copy sets of pulses from one state to other multiple states in the Pulse configuration dialog. This can be useful for quickly setting up a number of similar pulse protocols


keeps track of the current time. You can insert your own code for on-line analysis
into the idle routine in this example.

Scripts: Signal

Q. Is it at all possible to control older Magstim 200 series models that don't have a
serial line connection? There is a modification available for these units that
introduces a parallel port which can be used to control the intensity settings so is
there any way to interface to this from the 1401?

A. The attached script, Ctr1200series.sgs, can be used to control older
Magstim 200 models that have been fitted with the parallel port modification,
using multiple states to control the 1401 digital outputs. The script creates a
suitable sampling configuration and allows the user to specify the Magstim output
intensities and interval between outputs for each state used. The connection
between the Magstim parallel port interface and the 1401 digital outputs can be
achieved with a modified cable (contact us for details).

Did you know...?

The automate tab in Spike2 and Signal sampling configurations now contains options for
generating file names for data files that include the data and time at which the file was
created.

Sampling configuration - C:'\Spikeb ' LAST.SZC* x|

Ehannelsl Hesulutionl tode I Sequencerl Play waveform  Automation |

—Automatic file name generation

Directary path |C:'\SpikeEKdata\ Browse. .. | ‘

wymmdd_000: 080304 _000
ddmrey_ 000;  040308_000

Mame termplate ID ata_00

~ Data security Staing  Month (01-12): 03
Flush to disk evem |2 rrifbes Trigger  ‘ear (00-99%: 08
¥ Comment file at end of zampling ¥ St EETE 2008

Hour {00-237: 15
¥ Savefile at end of sampling [~ sk Minute (00-59% 52
— Sample bar Second (00-590: 21

Label I LComment I

Ok | Cancel Run now Help

Spike2 sampling configuration with file name template options

Recent questions

Q. | would like to be able employ reusable functions in Spike2 scripts. Is there a way
for scripts to call functions stored in other scripts without having to copy and
paste them into every script that needs them?

A. The script language in both Spike2 and Signal support #include statements.
This allows the user to include existing script files that contain definitions or user-
defined functions and procedures into a script, rather than copying and pasting
sections of code from one script to another. See the on-line help for more
information on using the #include statements in scripts.

User group

If you have any comments about the newsletter format and content, or wish to un-
subscribe from the mailing list for this newsletter, please notify sales@ced.co.uk.

All Trademarks are acknowledged to be the Trademarks of the registered holders.

Back to top

Right-click the script icon
and save to disk.

If you have any problems
opening the embedded
scripts in this newsletter
please let us know.

'$Ctrl200series.sgs|Example script to provide basic control over Magstim 200 series models that have been

'fitted with the parallel port interface option.



'The script is a 'work in progress' and is offered without guarantees.

'You must test it to see whether it is suitable for your application.



'CED 10/03/08



var data%;                      'View handle of data file

var sRate% := 5000;             'Sample rate of waveform channels

var fLength := 500;             'Frame length (ms);

var chargeTime := 4;            'Additional outputs frame time for charging Magstim

var sPts%;                      'Data points per frame

var sOffset% := 100;            'Pre trigger time (ms)

var sOffset;                    'Pre trigger time in seconds

var nStates% := 5;              'Number of additional states

var int%[20];                   'Array of Magstim intensity levels (one per state)

var label$[20];                 'Array of state labels  

var FrameInt[20];               'Array in intervals to next frame

var TrigDir% := 1;              'Trigger polarity



ArrConst(frameInt[],5);         'Initialise frame interval



ToolbarSet(1, "Quit");          'Toolbar for script control

ToolbarSet(3, "Settings", GetSet%);

ToolbarSet(4, "New file", New%);

ToolbarEnable(4,0);

Toolbar("TMS configuration setup",1023);





Func GetSet%();

var ok%;

var okState%;

DlgCreate("Sampling parameters");               'Set sampling parameters 

DlgInteger(1, "Sample rate",100,100000);

DlgReal(2, "Frame length(ms)",50,10000);

DlgInteger(3, "Pre-trigger time",0,500);

DlgReal(4, "Charge time",0,10);

DlgInteger(5, "Number of states",1,20);

DlgList(6, "Trigger polarity","Rising|Falling",2);

ok% := DlgShow(sRate%,fLength,sOffset%,chargeTime,nStates%,trigDir%);           'Assign return values to variables

if ok% > 0 then                                                                 'If OK pressed        

    sPts% := (fLength/1000.0)* sRate%;                                          'Get number of sample points per frame

    sOffset := sOffset% / 1000.0;                                               'Convert offset to seconds

    okState% := GetStateParams%(NStates%);                                      'Jump to function to get individual state settings

    if okState% > 0 then

        SetConfig%();                                                           'Set up configuration

        ToolbarEnable(4,1);

    endif;

endif;

return 1;

end;





'Function to get & set individual state parameters (Magstim intensity, state label and interval to next frame)

Func GetStateParams%(n%);                               

var ok%;

var i%;

DlgCreate("State setup");       'Setup Magstim settings for multiple states

DlgText("State",0,1);

DlgText("Intensity",12,1);

DlgText("Label",26,1);

DlgText("Interval(s)",36,1);

for i% := 1 to n% do

    DlgText(Str$(i%),3,i%+1);

    DlgInteger(i%,3,1,127,14,i%+1);

    DlgString(n%+i%,"",10,"",24,i%+1);

    DlgReal((n%*2)+i%,5,chargeTime + 0.1,3600,40,i%+1);

next;

ok% := DlgShow(int%[:nStates%],label$[:nStates%],frameInt[:nStates%]);

if ok% > 0 then

    if trigDir% = 1 then

        ArrAdd(int%[:nStates%],128);        

    endif;

endif;

return ok%;

end;





Func New%()

data% := FileNew(0,0);          'Open new file for sampling

if data% > 0 then

    Window(0,0,100,100);                

    FrontView(data%);

    ToolbarClear(4);

    ToolbarSet(4, "Start",Start%);

endif;

return 1;

end;





Func Start%()

View(data%);

SampleStart();      'Start sampling

ToolbarClear(4);

ToolbarSet(4, "Stop",Stop%);

return 1;

end;





Func Stop%()

View(data%);

SampleStop();       'Stop sampling

ToolbarClear(4);

ToolbarSet(4, "New file", New%);

return 1

end;





Func SetConfig%()

var i%;

SampleClear();                                      'Clear existing configuration

SampleBurst(1);                                     'Burst mode sampling

SamplePorts(2);                                     'Sample 2 ports

SamplePortUnits$(0," Volt");                        

SamplePortFull(0,5);

SamplePortZero(0,0);

SamplePortName$(0,"ADC 0");

SamplePortUnits$(1," Volt");

SamplePortFull(1,5);

SamplePortZero(1,0);

SamplePortName$(1,"ADC 1");

SampleKeyMark(1);                                   'Enable keyboard marker

SampleDigMark(0);                                   'Disable digital marker

SampleAutoName$("");                                'No auto file naming

SampleAutoFile(0);              

SamplePoints(sPts%);                                'Number of data points per ADC port per frame

SampleRate(sRate%);                                 'Sample rate

SampleZeroOffset(sOffset);                          'Pre zero time

SampleTrigger(1);                                   'Turn on sweep trigger

SampleMode(3);                                      'Sample mode to fixed interval

SampleOutMode(1);                                   'Pulses outut mode

SampleDigOMask(255);                                'Enable all 8 digital outputs

SampleOutClock(0.001);                              'Set output resolution to 1ms

SampleStates(nStates%);                             'Number of states

SampleStatesMode(2);                                'Dynamic outputs state mode

SampleStatesOrder(0);                               'Numeric state ordering

SampleStatesIdle(0);

SampleStatesRepeats(1);

SampleStatesOptions(2);

SamplePause(0);

SampleWrite(1);

PulseClear(0,-1);                                                  'Clear all pulses

if trigDir% = 0 then

    PulseDataSet(0,-1,0,0);

else

    PulseDataSet(0,-1,0,128);

endif;

for i% := 1 to nStates% do                                              'Set up individual state parameters

    PulseClear(i%,-1); 

    SampleOutLength(i%,chargeTime+(fLength/1000.0));                    'Set output length to same as frame length

    SampleOutTrig(i%,chargeTime);                                       'Set outputs frame to include charge time

    PulseDataSet(i%,-1,0,int%[i%-1]);                                   'Set bit value to intensity level 

    PulseAdd(i%,-1,1,"Stimulus",chargeTime + sOffset,0.01);             'Add stimulus trigger pulse at time zero

    PulseDataSet(i%,-1,"Stimulus",128);                             'Set it to be on digital output 7 

    SampleStateLabel$(i%,label$[i%-1]);                                 'Set state label

    SampleFixedInt(i%,frameInt[i%-1]);                                  'Set interval to next frame

next;

return 1;

end;







CED
Ctrl200series.sgs

CED
The last sampling configuration to be used is always saved as Last.S2C and will be the default configuration the next time you sample. You can load and save your own specific configurations using the Load and Save configuration commands in the File menu




